Euclidea: Algunas soluciones y demostraciones

Luis Mochán

13 de abril de 2026

1. Introducción

Euclidea es una colección de problemas geométricos de dificultad creciente, con diversas interfaces, incluyendo una app para Android. De repente me he atorado en algunos problemillas y he hallado soluciones a veces de manera accidental, o por simple insistencia, o usando criterios heurísticos. En algunas de esas ocasiones me ha costado trabajo encontrar o visualizar las demostraciones de que mi solución es correcta. En Euclidea todos los círculos son completos y todas las líneas son infinitas, por lo que los diagramas se saturan rápidamente. Además, no hay un mecanismos para rotular las líneas. Mis dibujos a mano son muy malos. Esto dificulta hacer y visualizar demostraciones. Se me ocurrió usar `metapost` para generar diagramas más bonitos y limpios. Este documento fue generado usando el modo `org-mode` del sistema `emacs`, y en particular, su implementación de `noweb` para hacer *programación culta* (*literate programming*), mezclando código y texto. Los detalles del código se muestran hasta el final.

2. Problemas

2.1. β 4

El problema es hallar una tangente a un círculo que pase por un punto dados dicho punto y el centro.

Dibujamos el círculo inicial, con centro en 0 y buscamos la tangente en p.

```
<<header>>
  beginfig(1)
  <<headerfig>>
  <<primer círculo>>
```

```
endfig;
```



Luego, colocamos un punto arbitrario q en el círculo.

```
beginfig(2)
<<headerfig>>
<<primer círculo>>
<<punto arbitrario>>
endfig;
```



y con centro en dicho punto trazamos otro círculo que pase por p,

```
beginfig(3)
<<headerfig>>
<<primer círculo>>
<<punto arbitrario>>
<<segundo círculo>>
endfig;
```



Este círculo intersecta al círculo original en un punto r. Trazamos ahora un círculo centrado en p y de radio pr,

```
beginfig(4)
<<headerfig>>
<<primer círculo>>
```

```
<<punto arbitrario>>
<<segundo círculo>>
<<tercer círculo>>
endfig;
```



Este círculo intersecta al segundo círculo en un segundo punto s. Finalmente, la línea ps es tangente al primer círculo.

```
beginfig(5)
<<headerfig>>
<<primer círculo>>
<<punto arbitrario>>
<<segundo círculo>>
<<tercer círculo>>
<<tangente>>
endfig;
```



2.1.1. Demostración

Trazamos el radio del primer círculo, $0p$ y las líneas pr, pq, ps, qr, qs, rs, y lo ángulos α ($p0q$), β (prq), γ (rpq) y $\$\delta$ (qps). Como p y q están en un círculo centrado en 0, y r es otro punto en la periferia de dicho círculo, entonces $\beta = \alpha/2$. Como además, p y r están también en un círculo centrado en q, entonces el triángulo rqp es isóceles. Luego, $\gamma = \beta$ y $\gamma = \alpha/2$. Notamos que los lados del triángulo sqp son iguales a los lados del triángulo rqp, ambos triángulos son congruentes, y por tanto $\delta = \gamma$, luego $\delta = \alpha/2$. Finalmente, el triángulo $q0p$ es isóceles, y por tanto, el ángulo $\phi = \epsilon + \gamma$ ($0qp$) es $(\pi - \alpha)/2$ y $\$\epsilon{=}\pi/2 - \alpha$. Por lo tanto, el ángulo entre el radio $0p$ y la línea ps es $\epsilon + \gamma + \delta = \pi/2$, i.e., la línea ps es ortogonal al radio del primer círculo y por lo tanto es tangente en el punto p.

```
beginfig(6)
<<headerfig>>
```

```
<<líneas>>
endfig;
```



2.1.2. Códigos metapost

```
g_use_svg = 1; %0 for postscript, 1 for svg
prologues:=3;
u=1cm;
outputtemplate := "%j-%c.eps";
if g_use_svg > 0:
        outputtemplate := "%j-%c.svg";
```

```
        outputformat := "svg";
fi
```

Figura 1: Encabezado global «header»

```
save r, a, p;
numeric r[];
pair a[];
path p[];
r0=10u; % radio primer círculo
a0=point 1 of fullcircle scaled r0;
a1=point 2.2 of fullcircle scaled r0;
p0=fullcircle scaled r0;
p1=fullcircle scaled (2*abs(a0-a1)) shifted a1;
a2=p1 intersectionpoint p0;
p2=fullcircle scaled (2*abs(a2-a0)) shifted a0;
a3=p2 intersectionpoint p1;
```

Figura 2: Encabezado «headerfigs»

```
draw p0;
fill fullcircle scaled 10pt;
fill fullcircle scaled 10pt shifted a0;
label.lrt(btex $0$ etex scaled 2,(0,0));
label.lrt(btex $p$ etex scaled 2, 1.02a0);
```

Figura 3: «primer círculo»

```
fill fullcircle scaled 10pt shifted a1;
label.urt(btex $q$ etex scaled 2, 1.02a1);
```

Figura 4: «punto arbitrario»

```
draw p1;
fill fullcircle scaled 10pt shifted 1a2;
label.ulft(btex $r$ etex scaled 2, 1.02a2);
```

Figura 5: «segundo círculo»

```
draw p2;
fill fullcircle scaled 10pt shifted a3;
label.ulft(btex $s$ etex scaled 2, 1.02a3);
```

Figura 6: «tercer círculo»

```
draw a0--a3;
```

Figura 7: «tangente»

```
draw (0,0)--a0;
draw (0,0)--a1;
draw a0--a1;
draw a0--a2;
draw a0--a3;
draw a1--a2;
draw a1--a3;
draw a2--a3;
fill fullcircle scaled 10pt;
fill fullcircle scaled 10pt shifted a0;
label.lrt(btex $0$ etex scaled 2,(0,0));
label.lrt(btex $p$ etex scaled 2, 1.02a0);
fill fullcircle scaled 10pt shifted a1;
label.urt(btex $q$ etex scaled 2, 1.02a1);
fill fullcircle scaled 10pt shifted 1a2;
label.ulft(btex $r$ etex scaled 2, 1.02a2);
fill fullcircle scaled 10pt shifted a3;
label.ulft(btex $s$ etex scaled 2, 1.02a3);
label(btex $\alpha$ etex scaled 2, unitvector(a0+a1)*1.1u);
label(btex $\beta$ etex scaled 2, a2+unitvector(a0+a1-2a2)*1.5u);
label(btex $\gamma$ etex scaled 2, a0+unitvector(a1+a2-2a0)*1.5u);
label(btex $\delta$ etex scaled 2, a0+unitvector(a1+a3-2a0)*1.5u);
label(btex $\epsilon$ etex scaled 2, a0+unitvector(a2-2a0)*1.5u);
label(btex $\phi$ etex scaled 2, a1+unitvector(a0-2a1)*1.1u);
```

Figura 8: «líneas»

```
.end
```

Figura 9: Fin